
Введение.

Подготовка к использованию.

Чтобы иметь возможность использовать класс «ПрямойЗапрос» в 1С: Предприятие 7.7 нам понадобится внести несколько несложных изменений в программу.

Для этого нам понадобится:

1. Внешняя компонента 1С++. Приложена в архиве.

2. Внешняя компонента 1sqlite. Приложена в архиве.

3. Класс «ПрямойЗапрос» (реализован в виде внешней обработки ПрямойЗапрос.ert)
4. Файл инициализации класса defcls.prm. Вместе с классом в папке ПрямыеЗапросы в архиве.

Подключение внешних компонент необходимо произвести в процедуре ПриНачалеРаботыСистемы() в глобальном модуле конфигурации. Данные действия реализуются стандартным образом. А именно посредством команды ЗагрузитьВнешнююКомпоненту().

Далее необходимо создать обработку с идентификатором Defcls. Данная обработка необходима для инициализации класса «ПрямойЗапрос». В текст обработки необходимо поместить следующий текст:
//#include [ПутьККаталогуКласса]\Defcls.prm
Где [ПутьККаталогуКласса] – путь к каталогу класса относительно каталога информационной базы. Например: он может выглядеть следующим образом: SystemData\ПрямыеЗапросы\Defcls.prm. В данном примере SystemData – каталог, расположенный в каталоге ИБ, а ПрямыеЗапросы – каталог, расположенный в каталоге SystemData, в котором находится файл инициализации класса (п.4) и собственно сам класс (п.3).

После этого можно запустить данную конфигурацию в режиме «Предприятие». После чего необходимо проверить подключился ли данный класс. Для этого необходимо зайти в меню Сервис -> Параметры. Если в списке закладок, открывшегося окна, есть закладка Настройки 1С++, значит загрузка компоненты 1С++ прошла успешно. 

Перейдем на данную закладку. На ней расположено окно со списком зарегистрированных классов. А также интересующая нас кнопка «Оптимизация». Отключенная оптимизация в основном необходима для отладки класса в процессе разработки и для конечного пользователя не имеет никакого значения, кроме влияния на время создания экземпляра класса. При включенной оптимизации время создания первого объекта может быть 600-1000 мс, последующие объекты будут создаваться около 6-12 мс.
В общем если в списке зарегистрированных классов присутствует наш класс и кнопка Оптимизация включена, то можно приступать к работе.

Таблицы, поля базы данных.
Таблицы в «1С: Предприятие 7.7» подразделяются на три основных типа: реальные, виртуальные и временные.

Реальные таблицы хранятся в базе данных. В случае использования реальной таблицы могут присутствовать вычисляемые поля, значения которых вычисляются как функции нескольких разных полей.

Виртуальные таблицы в базе не хранятся. При обращении к информации виртуальных таблиц система автоматически собирает информацию реальных таблиц для выполнения запроса. Виртуальные таблицы являются параметризированными.

Временные таблицы создаются пользователем самостоятельно, для хранения тех или иных данных на время выполнения запроса.

В свою очередь любая таблица состоит из набора полей. Поле может содержать значение одного типа, или значения разных типов, при этом для конкретной записи поле содержит значение «одного» типа.

В качестве простых типов могут выступать:

· Число

· Строка

· Дата

· Null
Помимо этих типов в качестве типов полей могут выступать типы, появившиеся после их определения в конфигураторе.
Структура запроса (описание запроса).
В качестве механизма получения данных в классе используется механизм запросов. Запросы получают необходимые данные из таблиц информационной базы. Для того чтобы запрос получил «нужные» данные, их состав необходимо «указать». Указание производится на специальном языке запросов, поддерживаемым объектом «ПрямойЗапрос». С помощью конструкции данного языка составляется текст запроса.

Можно сказать, что текст запроса – это инструкция, в соответствии с которой должен быть выполнен запрос. Данный текст имеет определенную структуру:

ВЫБРАТЬ [РАЗЛИЧНЫЕ] [ПЕРВЫЕ|TOP <Количество>]

<Список полей выборки>

ИЗ <Список источников>

[ГДЕ <Условие отбора>]

[СГРУППИРОВАТЬ <Поля группировки>]

[ИМЕЮЩИЕ <Условие отбора>]

[УПОРЯДОЧИТЬ <Условия упорядочивания>]
[LIMIT <Количество>]

В минимальном варианте конструкция может содержать только конструкцию «ВЫБРАТЬ … ИЗ …».

После составления текста запроса его необходимо передать объекту «ПрямойЗапрос» для выполнения. Результатом выполнения будут объекты: ИндексированнаяТаблица, СписокЗначений или ТаблицаЗначений.
Итак, подведем итоги данной главы:

1. Для работы класса необходимо наличие внешних компонент 1С++ и 1sqlite.
2. Наличие зарегистрированного в конфигурации класса «ПрямойЗапрос».
3. Для получения данных используется специальная инструкция – запрос, который описывается на специальном языке запросов.

4. Текст запроса имеют четкую структуру.

5. Результатом выполнения запроса будут объекты: ИндексированнаяТаблица, СписокЗначений или ТаблицаЗначений.

Описание механизмов объекта «ПрямойЗапрос».

Атрибуты и методы объекта «ПрямойЗапрос».

Подробно все атрибуты и методы описаны в документации к классу. Поэтому здесь мы будем говорить о них по мере необходимости.

Для начала нам понадобятся атрибуты:

· Текст

· РежимОтладки

И методы:

· Выполнить

· УстановитьТекстовыйПараметр

Атрибут «Текст» хранит текст запроса, написанный пользователем класса, т.е. программистом (далее «Пользователь»). Синтаксис языка запроса подробно изложен в документации. Мы же будем разбираться с ним постепенно.

Атрибут «РежимОтладки» может принимать значения 0 или 1 (на самом деле их больше, но в рамках данной инструкции это не имеет значение). В случае если РежимОтладки = 1 в процессе выполнения запроса объектом будет показан реальный текст запроса который уходит на выполнение. При этом для DBF формата БД также будет показано использование индексов при запросах к таблице и их стоимость. Т.о. можно будет сделать вывод об оптимальности выполнения запросов. Впрочем об этом чуть ниже.

Метод «Выполнить» предназначен для обработки текст запроса установленного пользователем в атрибуте Текст, выполнении подготовленного запроса и возвращению результата выполнения в виде объекта ТаблицаЗначений, СписокЗначений или ИндексированнаяТаблица. Метод имеет два атрибута (необходимых для работы).

· ТипОбъекта. Тип: Строка. Тип объекта для возвращения результата запроса. Возможные значения:

· ИндексированнаяТаблица (по умолчанию).
· ТаблицаЗначений

· СписокЗначений

· ТекстЗапроса. Тип: Строка. Текст запроса для выполнения. При этом текст установленный в атрибуте «Текст» не изменяется, а выполняется текст запроса переданный в данном параметре.
Метод «УстановитьТекстовыйПараметр» предназначен для установки значений пользовательских параметров указанных в тексте запроса. О параметрах и их использовании поговорим ниже. Метод имеет два атрибута:

· ИдентификаторПараметра. Тип: Строка. Идентификатор пользовательского параметра.

· ЗначениеПараметра. Тип: Число, Строка, Дата и объекты 1С: Предприятие. Значение для установки в текст запроса.

Итак: на первом этапе работы нам необходимо запомнить всего 1 атрибут – Текст, и 1 метод – Выполнить. Пример:

ПрямойЗапрос = СоздатьОбъект(”ПрямойЗапрос”);

ПрямойЗапрос.Текст = ”ВЫБРАТЬ

|
$Спр.ТекущийЭлемент 
|
,$Спр.Наименование
|ИЗ 
|
Справочник.Контрагенты КАК Спр $nolock”;

ТаблицаЗначений = ПрямойЗапрос.Выполнить(”ТаблицаЗначений”);
Итак, подведем итоги данной главы:

1. Основной текст запроса записывается в атрибут «Текст».
2. Выполнение запроса с возвращением результата осуществляется методом «Выполнить()».

3. Для более подробной информации о ходе выполнения запроса используется РежимОтладки = 1.

Первый запрос.

Вот мы и увидели наш первый запрос.
ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК ТекущийЭлемент

,$Спр.Наименование КАК Наименование

,$Спр.ОсновнойДоговор КАК ОсновнойДоговор
ИЗ 


Справочник.Контрагенты КАК Спр $nolock
Давайте разберемся, как это работает. Для этого разделим запрос на две части:

1. Запрашиваемые колонки

2. Источники данных

Запрашиваемые колонки – это список колонок, которые мы хотим иметь в результирующей таблице. Источники данных – это указание таблиц, из которых мы хотим получить данные.

В нашем примере запрашиваемые колонки: ТекущийЭлемент, Наименование и ОсновнойДоговор.
Источник данных: объект конфигурации справочник «Контрагенты».

Обратите внимание на то, что необходимые нам колонки мы указывали при помощи специального символа «Алиас» - $, и с указанием псевдонима таблицы – источника для данной колонки.А также добавляли имя представления колонки в результате выполнения запроса («Колока КАК Представление»).
Данная особенность в первую очередь связана с тем как 1С: Предприятие хранит данные в программе. Если не вдаваться с подробности то, все создаваемые реквизиты у справочников, документов, счетов, проводок, операций, а также измерения и ресурсы у регистров хранятся во внутренних представлениях, а именно как колонки spXXX, где XXX – число. Кроме этого существуют служебные поля, такие как code, descr, ismark, isfolder, iddoc, id и пр. 
Более того, все таблицы объектов: справочников, документов, регистров, проводок и т.д. хранятся также во внутренних представлениях. Например: таблицы справочников – это таблицы scXXX, таблицы реквизитов шапки документов – это dhXXX, реквизиты многострочной части документов – это dtXXX и т.д.
Для начала разберемся, как нам указывать источники данных.

Источники данных делятся на три вида: 

1. Таблицы объектов метаданных конфигурации (справочники, документы, многострочные части документов)

Данные таблицы указываются в источниках «как есть», т.е. Справочник.Контрагенты, Документ.СчетНаОплату, ЖурналДокументов и т.д. Для данных таблиц обязательно указание представления таблицы в запросе, т.е. ее переименование. Например: Справочник.Контрагенты КАК Спр. При этом после всех таблиц обязательно указание специального слова $nolock. 

Колонки данных таблиц указываются по правилу $ПредставлениеТаблицы.ИдентификаторКолонки, где: $ - обязательный символ, именно по нему ориентируется обработки текстов запроса при подготовке; ПредставлениеТаблицы – представление таблицы указанное в источнике, по данному представлению обработчик ориентируется какой объект данных представлен перед ним; ИдентификаторКолонки – идентификатор предопределенного или созданного реквизита.

2. Виртуальные таблицы (данные таблицы преобразуются во вложенный подзапрос)

Для данных таблиц обязательно указание специального символа «Алиас» - $ перед идентификатором таблицы. Также как и для таблиц объектов метаданных для данных таблиц обязательно наличие представления таблицы в запросе. Например: $РегистрОстатки.ОстаткиТМЦ(,,,) КАК ОстаткиТМЦ.

Колонки данных таблиц указываются по правилу ПредставлениеТаблицы.ИдентификаторКолонки, где ПредставлениеТаблицы – представление таблицы указанное в источнике; ИдентификаторКолонки – идентификатор колонок виртуальной таблицы. Обратите внимание, что виртуальная таблица переименовывает поля внутри себя, поэтому указывать спец. символ - $ нет необходимости.

3. Временные таблицы (таблицы созданные пользователем для выполнения различных задач).

Для данных таблиц обязательно указание специального символа «#» перед идентификатором таблицы. Также как и для таблиц объектов метаданных для данных таблиц обязательно наличие представления таблицы в запросе. Например: #ВременнаяТаблицаКонтрагенты КАК ТабКонтрагенты.

Колонки данных таблиц указываются по правилу ПредставлениеТаблицы.ИдентификаторКолонки, где ПредставлениеТаблицы – представление таблицы указанное в источнике; ИдентификаторКолонки – идентификатор колонок временной таблицы. Обратите внимание, что как и для виртуальной таблицы указывать спец. символ - $ нет необходимости.

Среди таблиц объектов метаданных конфигурации существуют следующие:

· Справочник.<ВидСправочника> (scXXX)

· Документ.<ВидДокумента> (dhXXX)
· ДокументСтроки.<ВидДокумента> (dtXXX)
· ЖурналДокументов (1sjourn)
· Регистр.<ВидРегистра> (raXXX)
· РегистрИтоги.<ВидРегистра> (rgXXX)
· СсылкиДокументов (1scrdoc)
· Константы (тут же и периодические реквизиты) (1sconst)
· ПланыСчетов (1saccs)
· Операции (1soper)
· Проводки (1sentry)
· Обороты (1sbkttlc)
· Остатки (1sbkttl)
· ОтборСчетов (1saccsel)
· ОтборПроводок (1ssbsel)
Среди виртуальных таблиц существуют следующие:

· $ПодчиненныеДокументы.<ВидПодчинения>
· $СрезПервых.<ВидСправочника>
· $СрезПоследних.<ВидСправочника>
· $РегистрОстатки.<ИмяРегистра>
· $РегистрОбороты.<ИмяРегистра>
· $РегистрОстаткиОбороты.<ИмяРегистра>
· $БИОстатки.<ВидПланаСчетов>
· $БИОборотыДтКт.<ВидПланаСчетов>
· $БИОбороты.<ВидПланаСчетов>
· $БИОстаткиОбороты.<ВидПланаСчетов>
Более подробное описание источников данных и особенностей работы с ними разобраны в соответствующих разделах данной инструкции.

Таким образом, если бы мы выполняли запрос без использования данного класса, то он бы выглядел примерно вот так: 

SELECT

Спр.Id as ТекущийЭлемент

,Спр.descr as Наименование

,Спр.sp13 as ОсновнойДоговор
FROM 


sc133 as Спр (nolock)
Согласитесь, что понимание и написание подобного запроса было бы крайне неудобно и более того, отнимало бы значительно больше времени.
Поэтому в объекте «ПрямойЗапрос» реализован механизм подготовки текстов запроса, для исполнения их исполнения в DBF или SQL формате базы данных.

Теперь вернемся к нашему запросу. 
ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК ТекущийЭлемент

,$Спр.Наименование КАК Наименование


,$Спр.ОсновнойДоговор КАК ОсновнойДоговор
ИЗ 


Справочник.Контрагенты КАК Спр $nolock
После выполнения запроса мы получим таблицу:

	ТекущийЭлемент
	Наименование
	ОсновнойДоговор

	   3EV   
	ООО «Рога и копыта»
	     1   

	   3EW   
	ООО «Белочка»
	     A   

	   3EX   
	ООО «Мельница»
	     0   


Если посмотрим на таблицу, то поймем что вместо элементов справочника «Контрагенты» и «Договоры» мы получили какие-то строковые значения. Это внутренние идентификаторы данных объектов 1С: Предприятие. Их преобразование в значение удобное для использования можно возложить на плечи пользователя в момент обработки результата запроса, но удобнее и правильнее это делать средствами запроса. Для этого существует специальные правила. Подробнее о типизации конкретных значений к типам и видам 1С можно посмотреть в документации к классу. А в качестве примера приведем наш запрос к такому виду, чтобы типизация выполнялась до возвращения результата запроса.
ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]


,$Спр.Наименование КАК [Наименование $Строка]


,$Спр.ОсновнойДоговор КАК [ОсновнойДоговор $Справочник.Договоры]

ИЗ 


Справочник.Контрагенты КАК Спр $nolock
После выполнения запроса мы получим таблицу:

	ТекущийЭлемент
	Наименование
	ОсновнойДоговор

	ООО «Рога и копыта»
	ООО «Рога и копыта»
	№ 1 от 10.09.09

	ООО «Белочка»
	ООО «Белочка»
	№ 101 от 12.09.09

	ООО «Мельница»
	ООО «Мельница»
	


В результате значения в колонке ТекущийЭлемент и ОсновнойДоговор будут типизированы и возвращены в виде элементов соответствующих справочников. Важно помнить, что возвращаются ссылки, а не объекты.

Итак, подведем итоги данной главы:

1. Все таблицы являющиеся объектами 1С: Предприятие объявляются в запросе «как есть» с обязательным спец. словом $nolock в конце. 
2. Виртуальные таблицы используются со спец. символом $.

3. Временные таблицы пользователя со спец. символом #.

4. Поля данных являющихся полями таблиц объектов 1С: Предприятие должны объявляться со спец. символом $ и указанием псевдонима таблицы источника.

5. Все объявленные поля запроса необходимо переименовывать. Иначе совместимость между DBF и SQL форматами данных не гарантируется.

6. Для типизации значений в типам и видам 1С: Предприятие используется конструкция [ИмяПоля $ТипВидДанных]. Где ИмяПоля – переименованное имя поля которое возвращается в виде имени колонок в таблице, ТипВидДанных – строковая конструкция с указанием типа и вида данных для приведения значений. Приведение производится к типам указанным в конфигураторе.
Более подробное описание языка запросов, а также его возможностей можно найти в документации к классу. Далее мы продолжим изучать возможности запросов с приложением к конкретным объектам 1С: Предприятие.
Запросы к справочникам.

На каждый справочник, определенный в конфигураторе существует таблица. Имя таблицы складывается из ключевого слова «Справочник» и через точку добавляется идентификатор справочника, заданный в конфигураторе. Например: «Справочник.Номенклатура».
Таблица справочника может обладать следующим набором полей:

	Справочник.<Идентификатор>
	Тип значения

	Код
	Число, Строка

	Наименование
	Строка

	ПометкаУдаления
	Число

	ЭтоГруппа
	Число

	Родитель
	Справочник.<Идентификатор>

	Владелец
	Справочник.<ИдентификаторВладельца>

	ТекущийЭлемент
	Справочник.<Идентификатор>

	Реквизит
	Тип реквизита


Описание полей выборки, источники в запросе.
Просмотреть элементы справочника можно с помощью простого запроса:

ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]

ИЗ 


Справочник.Контрагенты КАК Спр $nolock
После обязательного слова «ВЫБРАТЬ» следует описание полей запроса. Описание полей запроса содержит указание поля таблицы запроса, имя результирующего поля, указание типизации значения. Обратим внимание, поскольку мы обращаемся к реальной таблице базы данных имя реквизита указывается при помощи специального символа – Алиас - «$», с указанием псевдонима таблицы источника. Имя результирующего поля и типизация значения указаны в квадратных скобках «[]». Ключевое слово «КАК» не является обязательным для выполнения запроса.
Источник запроса (под источником обычно понимается таблица, реальная, виртуальная или временная) указывается после ключевого слова «ИЗ». Строка источника запроса содержит: имя таблицы базы данных, ее представление и ключевое слово «$nolock», обязательное для всех реальных таблиц базы данных.
Конструкции Различные, Первые N.
Приведем пример текста запроса для получения всех единиц измерения, используемых в справочнике «Номенклатура».
ВЫБРАТЬ


$Спр.ЕдиницаИзмерения КАК [ЕдиницаИзмерения $Справочник.ЕдиницыИзмерений]

ИЗ 


Справочник.Номенклатура КАК Спр $nolock
В результате данного запроса все некоторые единицы измерений будут повторяться. Для получения только уникальных значений можно использовать ключевое слово «РАЗЛИЧНЫЕ». Например:

ВЫБРАТЬ РАЗЛИЧНЫЕ


$Спр.ЕдиницаИзмерения КАК [ЕдиницаИзмерения $Справочник.ЕдиницыИзмерений]

ИЗ 


Справочник.Номенклатура КАК Спр $nolock
Для получения значения только первых нескольких позиций можно использовать конструкцию ПЕРВЫЕ N, или сочетание TOP N|LIMIT N.

Разница в использовании данных конструкций в том, что «ПЕРВЫЕ N» - не может применяться во вложенных запросах, поскольку это приведет к ошибке выполнения запроса. TOP N|LIMIT N – более универсальная конструкция, и может быть использована в любом месте. При этом ключевое выражение TOP N должно быть сразу после ключевого слова ВЫБРАТЬ, а выражение LIMIT N, сразу после окончания конструкции запроса.
Например: реализация ПЕРВЫЕ N.

ВЫБРАТЬ ПЕРВЫЕ 5


$Спр.ЕдиницаИзмерения КАК [ЕдиницаИзмерения $Справочник.ЕдиницыИзмерений]

ИЗ 


Справочник.Номенклатура КАК Спр $nolock
Например: реализация TOP N|LIMIT N.

ВЫБРАТЬ TOP 5


$Спр.ЕдиницаИзмерения КАК [ЕдиницаИзмерения $Справочник.ЕдиницыИзмерений]

ИЗ 


Справочник.Номенклатура КАК Спр $nolock 

LIMIT 5
Фильтрация результатов запроса (отборы).

Основным преимуществом использования запросов в том, что при наличии определенных критериев нам нет необходимости выбирать все данные для получения желаемого результата, а затем перебирать их и делать сравнение (как это реализовано в стандартных механизмах 1С: Предприятие 7.7). Все условия мы можем задать в самом тексте запроса. 

Например необходимо выбрать контрагентов из справочника по следующим критериям:

· Наименование содержит символы «ООО» в начале

· ТипКонтрагента = Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

· ОсновнойДоговор = не пустое значение

ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]

ИЗ 


Справочник.Контрагенты КАК Спр $nolock
ГДЕ


$Спр.Наименование ПОДОБНО ‘ООО%’
И


$Спр.ТипКонтрагента = :Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

И


$Спр.ОсновнойДоговор = :ПустойОсновнойДоговор
В данном запросе добавился раздел условий, который начинается со слова «ГДЕ». Рассмотрим эти условия.

· Наименование содержит символы «ООО» в начале

Сначала мы указали поле запроса, которое хотим подвергнуть определенному сравнению. Правила именования поля в условиях такие же, как и в секции ВЫБРАТЬ. Далее мы указали условие сравнения – ПОДОБНО. Все возможные условия сравнения указаны в документации к классу. После чего указали значение сравнения. В данном случае это ‘ООО%’, где % - специальный символ, означающий любое количество любых символов после ООО. 

Но не всегда есть возможность указать значение условия запроса непосредственно в тексте самого запроса. Для этого существует понятие – текстовый параметр. Текстовый параметр – это специальная конструкция запроса, необходимая для подстановки значений в текст запроса непосредственно в момент его подготовки перед выполнением. 

Текстовые параметры делятся на два типа:

1. Предопределенные текстовые параметры. Их указание носит директивный характер и самостоятельно обрабатывается при подготовке запроса. Их установка непосредственно пользователем невозможна. Например: :ВидСправочника.Контрагенты, :ВидДокумента.СчетФактураВыданный и пр. Подробнее см. в документации к классу «ПрямойЗапрос».

2. Пользовательские текстовые параметры. Значения пользовательских параметров устанавливаются при помощи метода «УстановитьТекстовыйПараметр» объекта класса «ПрямойЗапрос». Установка значений для подстановки в запрос обязательна. В случае если в тексте присутствует пользовательский текстовый параметр значение которого не установлено, подготовка запроса не будет выполнена.

Текстовые параметры указываются следующим образом: «:ИдентификаторПараметра». Где «:» - специальный символ обозначающий параметр запроса, «ИдентификаторПараметра» - идентификатор или значение параметра, который указывается в методе «УстановитьТекстовыйПараметр» для пользовательских текстовых параметров.

Теперь вернемся к нашему запросу, а именно ко второму условию.

· ТипКонтрагента = Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

Значение перечисления можно устанавливать при помощи пользовательского текстового параметра, а также в виде предопределенного текстового параметра, который обрабатывается при подготовке. В нашем запросе мы воспользовались возможностью указать параметр в качестве предопределенного. Т.о. при подготовке запроса будет подставлено внутреннее значение перечисления.

Теперь посмотрим на третье условие запроса.

· ОсновнойДоговор = не пустое значение

Пустое значение на самом деле можно подставить тоже при помощи предопределенного текстового параметра. НО: здесь указан пользовательский параметр, в целях демонстрации работы с пользовательскими параметрами. В данном случае весь текст модуля должен выглядеть следующим образом:

ПрямойЗапрос = СоздатьОбъект(“ПрямойЗапрос”);

ПрямойЗапрос.Текст = “

|ВЫБРАТЬ

|
$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]

|
,$Спр.Наименование КАК [Наименование $Строка]

|
,$Спр.ОсновнойДоговор КАК [ОсновнойДоговор $Справочник.Договоры]

|ИЗ 

|
Справочник.Контрагенты КАК Спр $nolock
|ГДЕ

|
$Спр.Наименование ПОДОБНО ‘ООО%’
|И

|
$Спр.ТипКонтрагента = :Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

|И

|
$Спр.ОсновнойДоговор = :ПустойОсновнойДоговор
|”;

ПрямойЗапрос.УстановитьТекстовыйПараметр(“ПустойОсновнойДоговор”,ПолучитьПустоеЗначение(“Справочник.Договоры”));
ТЗ = ПрямойЗапрос.Выполнить(“ТаблицаЗначений”);
Обратите внимание, что мы подставили значение типизированное к справочнику «Договоры». Связано это с тем, как 1С хранит пустые значения собственных объектов. 
Вернемся к нашему первому не типизированному результату запроса:

	ТекущийЭлемент
	Наименование
	ОсновнойДоговор

	   3EV   
	ООО «Рога и копыта»
	     1   

	   3EW   
	ООО «Белочка»
	     A   

	   3EX   
	ООО «Мельница»
	     0   


Обратите внимание на значение колонки ОсновнойДоговор в последней строчке. Значение равно ‘     0   ’. Вот это и есть пустое значение справочника «Договоры» или любого другого объекта (документа, перечисления, счета и т.д.). Поэтому если бы мы подставили значение не типизированное, а просто скажем 0, то запрос значение условия  стало бы равно 0, а не строке вида ‘     0   ’.

В общем, после выполнения нашего запроса мы получим таблицу:

	ТекущийЭлемент
	Наименование
	ОсновнойДоговор

	ООО «Мельница»
	ООО «Мельница»
	


Но в нашем примере, колонка ОсновнойДоговор имела тип «Справочник.Договоры». Т.е. с конкретным указанием типа и вида значения.

Теперь представим что в нашем справочнике еще есть поле «КонтактноеЛицо» которое имеет тип «Справочник» без указания вида. 1С: Предприятие хранит данные поля как строковые значения в которых присутствует не только значение конкретного справочника (строка 9 символов), но и вид этого справочника во внутреннем представлении (строка 4 символа). Т.о. значение данной колонки будут строки длинною 13 символов – где 4 – это вид, а 9 – это значение.

Для установки значения такого пользовательского параметра, в тексте запроса используются модификаторы – спец. символ «~» после непосредственно идентификатора параметра. В таком случае, при подготовке запроса будет подставлено значение справочника с указанием вида справочника.

Более подробно о преобразовании значений 1С в значения базы данных можно почитать в соответствующей главе документации к классу. В нашем же случае мы добавим 4-е условие в запрос, для демонстрации работы модификатора:

· КонтактноеЛицо равно элемент справочника физ. лица

ВЫБРАТЬ


$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]


,$Спр.Наименование КАК [Наименование $Строка]


,$Спр.ОсновнойДоговор КАК [ОсновнойДоговор $Справочник.Договоры]

ИЗ 


Справочник.Контрагенты КАК Спр $nolock
ГДЕ


$Спр.Наименование ПОДОБНО ‘ООО%’
И


$Спр.ТипКонтрагента = :Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

И


$Спр.ОсновнойДоговор = :ПустойОсновнойДоговор

И


$Спр.КонтактноеЛицо = :ЗначениеФизЛица~

В данном запросе значением параметра «ЗначениеФизЛица» может выступать любой элемент справочника «ФизическиеЛица».

До этого мы использовали сравнения только с конкретными значениями, но это не всегда соответствует реальным условиям использования запросов.

Например, нам необходимо отбор контрагентов, у которых ОсновнойДоговор входит в список значений «ДоговораКонтрагентов». Конечно, мы можем написать условие сравнения с текстом «ИЛИ», но если значений в списке более 2-3 данной текст запроса будет очень большим и неудобным для чтения и понимания.

Для укладки списков в базу данных существует метод класса «ПрямойЗапрос» - «УложитьСписокЗначений» и метод «УложитьСписокОбъектов» (данные методы абсолютно идентичны). Подробнее об использовании данного метода можно прочитать в документации. 
В нашем же примере мы посмотрим, как можно это использовать.

ПрямойЗапрос = СоздатьОбъект(“ПрямойЗапрос”);

ПрямойЗапрос.Текст = “

|ВЫБРАТЬ

|
$Спр.ТекущийЭлемент КАК [ТекущийЭлемент $Справочник.Контрагенты]

|
,$Спр.Наименование КАК [Наименование $Строка]

|
,$Спр.ОсновнойДоговор КАК [ОсновнойДоговор $Справочник.Договоры]

|ИЗ 

|
Справочник.Контрагенты КАК Спр $nolock
|ГДЕ

|
$Спр.Наименование ПОДОБНО ‘ООО%’
|И

|
$Спр.ТипКонтрагента = :Перечисление.ТипыКонтрагентов.ЮридическоеЛицо

|И

|
$Спр.ОсновнойДоговор В (ВЫБРАТЬ val ИЗ #СписокДоговоров)
|”;

ПрямойЗапрос.УложитьСписокЗначений(СписокДоговоров,”#СписокДоговоров”,”Договоры”);
ТЗ = ПрямойЗапрос.Выполнить(“ТаблицаЗначений”);
Также существует более сложный метод «УложитьТаблицуЗначений». Но его использование мы рассматривать пока не будем.
Итак, подведем итоги данной главы:

1. Описание полей выборки должно содержать: указание поля данных, результирующее имя и типизацию значения.

2. Конструкция «ПЕРВЫЕ N» имеет ограничение на использование. Может применяться только в верхнем уровне запроса.

3. Поля для условий запроса при указании подчиняются тем же правилам, что и поля данных для выборки.
4. Для установки значений отборов используются текстовые параметры имеющие конструкцию «:ИдентификаторПараметра». Где: «:» - спец. символ обозначающий текстовый параметр, «ИдентификаторПараметра» - идентификатор пользовательского параметра указывающий на значение параметра установленного методом «УстановитьТекстовыйПараметр()».

5. Для приведения значений 1С: Предприятие к значениям базы данных используются модификаторы «~» или «*», которые указываются сразу же после идентификатора параметров без пробела. Символ «~» преобразует значение с указанием вида, символ «*» преобразует значение с указанием типа и вида.
4. Пустые значения для отбора обязательно должны быть типизированы.

5. Для укладки значений в таблицу в базе данных используется метод «УложитьСписокЗначений» или «УложитьСписокОбъектов».

Запросы к документам и многострочным частям.

Каждому документу, определенному в базе данных, соответствует таблица с именем, состоящим из ключевого слова «Документ.» и имени документа, опреденным в конфигураторе.
Данная таблица имеет следующий состав полей:

	Документ.<Идентификатор>
	Тип значения

	ТекущийДокумент
	Документ.<Идентификатор>

	<РеквизитШапки>
	Тип реквизита

	<ОбщийРеквизит> (Отбор = 0)
	Тип реквизита

	<РеквизитМногострочнойЧасти> (ИтогПоКолонке = 1)
	Число


Кроме ссылки на документ, реквизитов шапки конкретного вида документов, в данной таблицей доступны общие реквизиты без признака Отбор (данный признак задается в конфигураторе), а также итоговые значения по колонкам многострочной части с установленным признаком «ИтогПоКолонке» = 1. 
Многострочная часть документа хранится в отдельной таблице, которой соответствует имя состоящее из ключевого слова «ДокументСтроки.» и имени документа, определенным в конфигураторе.

Данная таблица имеет следующий состав полей:

	ДокументСтроки.<Идентификатор>
	Тип значения

	ТекущийДокумент
	Документ.<Идентификатор>

	НомерСтроки
	Число

	<РеквизитМногострочнойЧасти>
	Тип реквизита


Информация о номере, дате и времени документа, флаг проведения, пометка удаления и общие реквизиты с признаком отбор хранятся в таблице «ЖурналДокументов».

	ЖурналДокументов
	Тип значения

	ТекущийДокумент
	Документ.<Идентификатор>

	ВидДокумента
	ВидДокумента

	ПометкаУдаления
	Число (1,0)

	Проведен
	Число (1,0)

	ДатаДокумента
	Дата

	ВремяДокумента
	Строка

	НомерДокумента
	Строка

	Префикс
	Строка

	Регистр<ИдентификаторРегистра>
	Число (1,0)

	<ИдентификаторОбщегоРеквизита> (Отбор = 1)
	Тип реквизита


На самом деле в таблице «ЖурналДокументов» доступны еще и другие поля, но нам они ни к чему ибо носят больш служебный характер. Для получения значения данных полей необходимо использовать соединения данных.

Особенности таблицы ЖурналДокументов.

1. Поле ТекущийДокумент – хранит ссылку только на документ без указания вида. Т.о. по данным полям возможна связь между таблицами где существуют реквизиты с типом Документ и указанным видом документа. Внутреннее представление – Строка длинной 9 символов.

2. При запросе к журналу документов для получения типизированного поля ТекущийДокумент необходимо чтобы в тексте запроса обязательно присутствовало поле имеющее такое же имя (переименованное) с добавлением строки «_вид» в конце (внимание: чувствительно к регистру букв). Типизация данного поля ($ВидДокумента) не обязательна. Например: 

ВЫБРАТЬ


$Жур.ТекущийДокумент КАК [Документ $Документ]
,$Жур.ВидДокумента КАК [Документ_вид $ВидДокумента] 
ИЗ


ЖурналДокументов КАК Жур $nolock
3. Поле ВремяДокумента хранится в качестве 36-тиричного значения равного количеству милисекунд с начала дня до момента записи документа. Т.о. для типизации данного поля к значению вида ЧЧ:ММ:СС, необходимо при выборке из результата запроса произвести следующую манипуляцию:

ВнутреннееВремяДокумента = Результат.ВремяДокумента;

КоличествоМилисекунд = _StrToId(ВнутреннееВремяДокумента);

КоличествоСекунд = КоличествоМилисекунд/10000;

ЧЧ = Цел(КоличествоСекунд/3600);

ММ = Цел((КоличествоСекунд-ЧЧ*3600)/60);

СС = КоличествоСекунд – ЧЧ*3600 – ММ*60;

ВремяДокумента = Формат(ЧЧ,”Ч(0)2”)+”:”+Формат(ММ,”Ч(0)2”)+”:”+Формат(СС,”Ч(0)2”);
4. Для отбора документов между интервалом времени необходимо текущее значение времени перевести в количество милисекунд, после чего преобразовать методом _IdToStr и подставить в качестве параметра. Например:

ЧЧ = 0; ММ = 0; СС = 0;

ТекущееВремя(ЧЧ,ММ,СС);

КоличествоМилисекунд = (ЧЧ*3600+ММ*60+СС)*10000;

ВнутреннееВремяДокумента = _idToStr(КоличествоМилисекунд);

ПрямойЗапрос.УстановитьТекстовыйПараметр(«ВремяДокумента»,ВнутреннееВремяДокумента);
5. Поле ВидДокумента в зависимости от формата БД может иметь как строковое значение (DBF), так и числовое (SQL). Длина значения всегда = 4. Для отбора по данному полю используются предопределенный текстовый параметр «:ВидДокумента». 

6. Поле «Регистр<ИдентификаторРегистра>» хранит признак наличия движений по данному регистру. Количество таких полей равно количеству регистров объявленных в конфигурации.

Использование соединений.
Одним из способов указания взаимосвязи между таблицами является использование «Соединений».

Соединения бывают двух видов:

· Левое соединение (внешнее)

· Внутреннее соединение

На самом деле видов соединений несколько больше. Но в связи с ограничениями, наложенными на DBF формат базы данных, использование других соединений невозможно.

В любом случае, когда речь заходит о соединении, существует несколько связанных с этим понятий: Таблица № 1, Таблица № 2, соединение (его вид и условие).

Рассмотрим варианты на следующем примере:

Есть две таблицы:

	Автомобили
	
	Водители

	Наименование
	ГосНомер
	
	ГосНомерАвто
	ФИОВодителя

	Volvo S80
	У 001 ОА
	
	У 001 ОА
	Петров В.В.

	BMW 6er
	У 002 ОО
	
	У 002 ОО
	Иванов О.М.

	ВАЗ 2111
	У 003 ВТ
	
	У 005 КР
	Владимиров А.А.

	
	
	
	У 001 ОА
	Сидоров В.А.


Условием соединения будет: Автомобили.ГосНомер = Водители.ГосНомерАвто.

В результате запроса должны попасть данные:

	Наименование автомобиля
	ФИО водителя

	
	


В соответствии с условием можно выделить записи, для которых условие выполняется:

	Volvo S80
	У 001 ОА
	
	У 001 ОА
	Петров В.В.

	BMW 6er
	У 002 ОО
	
	У 002 ОО
	Иванов О.М.

	ВАЗ 2111
	У 003 ВТ
	
	У 005 КР
	Владимиров А.А.

	
	
	
	У 001 ОА
	Сидоров В.А.


Записи, не удовлетворяющие условию соединения: из таблицы № 1: ВАЗ 2111 – У 003 ВТ; из таблицы № 2 У 005 КР – Владимиров А.А. Как видно для записи в таблице № 1 нет парый в таблице № 2, как и наоборот.

Теперь рассмотрим варианты соединения:

Внутреннее соединение: в результат выполнения запроса войдут только данные записей для обеих таблиц, для которых выполняется условие соединения, т.е.

	Volvo S80
	Петров В.В.

	Volvo S80
	Сидоров В.А.

	BMW 6er
	Иванов О.М.


Левое соединение: в результат выполнения запроса войдут данные из записей, для которых выполняется условие соединения и «не вошедшие» из Таблицы № 1. Можно сказать, что в результат запроса войдут все данные из Таблица № 1, и для тех записей результата запроса, для которых выполялось условие соединения в полях, куда помещаются данные из Таблицы № 2, буду стоять значения, для которых условие не выполяется, будет стоять null.
	Volvo S80
	Петров В.В.

	Volvo S80
	Сидоров В.А.

	BMW 6er
	Иванов О.М.

	ВАЗ 2111
	Null


После рассмотрения абстрактного примера перейдем к более реальному.

ВЫБРАТЬ


$Док.Контрагент КАК [Контрагент $Справочник.Контрагенты]

,$Жур.НомерДокумента  КАК НомерДокумента


,$Жур.ДатаДокумента КАК [ДатаДокумента $Дата]

,$Жур.ТекущийДокумент КАК [Документ $Документ.Реализация]

ИЗ 


Документ.Реализация КАК Док $nolock
ЛЕВОЕ СОЕДИНЕНИЕ


ЖурналДокументов КАК Жур $nolock
ПО


$Док.ТекущийДокумент = $Жур.ТекущийДокумент
Группировки в запросе.
Приведем пример запроса получения информации о всех товарах и его количестве в проведенных документах «ПриходнаяНакладная».
ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,$ДокСтроки.Количество КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
ЛЕВОЕ СОЕДИНЕНИЕ 

ЖурналДокументов КАК Жур $nolock
ПО 

$ДокСтроки.ТекущийДокумент = $Жур.ТекущийДокумент

ГДЕ


$Жур.Проведен = 1

Результатом выполнения данного запроса будет таблица в которой товар будет повторяться столько раз сколько строк в разных документах было по нему заведено. Для получения ответа на вопрос «Сколько какого товара покупалось», потребуется свернуть результат запроса по товару с суммированием значений в колонке «Количество».

Данная операция может быть выполнена с помощью агрегатных функций:

ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,СУММА($ДокСтроки.Количество) КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
ЛЕВОЕ СОЕДИНЕНИЕ 

ЖурналДокументов КАК Жур $nolock
ПО 

$ДокСтроки.ТекущийДокумент = $Жур.ТекущийДокумент

ГДЕ


$Жур.Проведен = 1

СГРУППИРОВАТЬ


$ДокСтроки.Товар
Важно помнить, что в случае использования группировки все поля выборки должны делиться на агрегатные функции и поля, по которым ведется группировка.

Следующий запрос выполнится без ошибки:

ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,СУММА($ДокСтроки.Количество) КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
СГРУППИРОВАТЬ


$ДокСтроки.Товар
А попытка выполнить данный запрос приведет к ошибке:

ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,$ДокСтроки.ЕдиницаИзмерения КАК [ЕдИзм $Справочник.ЕдиницыИзмерений]


,СУММА($ДокСтроки.Количество) КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
СГРУППИРОВАТЬ


$ДокСтроки.Товар
При указании группировки псевдоним поля указывать нельзя.

В качестве агрегатных функций можно использовать:

· СУММА(Выражение)

· СРЕДНЕЕ(Выражение)

· МИНИМУМ(Выражение)

· МАКСИМУМ(Выражение)

· КОЛИЧЕСТВО([РАЗЛИЧНЫЕ] Выражение)

Агрегатные функции можно использовать и без определения группировок:

ВЫБРАТЬ


КОЛИЧЕСТВО($Спр.ТекущийЭлемент) КАК КоличествоКонтрагентов
ИЗ


Справочник.Контрагенты КАК Спр $nolock
Условия на значения агрегатных функций.
Если нам необходимо получить только товары чье общее количество прихода было более 1000, нам необходимо наложить выборку на агрегированное значение. Секция «ГДЕ» в данном случае не подходит, поскольку не умеет работать с агрегированными значениями. 

Рассмотрим пример следующего запроса:
ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,СУММА($ДокСтроки.Количество) КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
СГРУППИРОВАТЬ


$ДокСтроки.Товар

ИМЕЮЩИЕ


СУММА($ДокСтроки.Количество) > 1000

Следует помнить, что в конструкции «ИМЕЮЩИЕ» необходимо описывать условия только на агрегированные функции.

Упорядочивание результатов запроса.
Если посмотреть на результат запроса по товару (см. пример выше) то мы увидим что особого порядка там не прослеживается. Когда необходимо упорядочить записи по определенному правилу, то в данном случае может пригодится конструкция языка запросов «УПОРЯДОЧИТЬ».

ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,$ДокСтроки.Количество КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
УПОРЯДОЧИТЬ


$ДокСтроки.Товар ВОЗР
Возможные варианты упорядочивания: «ВОЗР» и «УБЫВ». В качестве поле упорядочивания можно указывать их псевдонимы.

Следует обратить внимание, что поскольку в поле Товар хранится не наименование, а ссылка на элемент справочника «Номенклатура» сортировка в данном случае будет произведена по значению ссылки, и может не совпадать с сортировкой по наименованию. Для сортировки по наименованию, необходимо выполнить соединение со справочником «Номенклатура» и отсортировать по полю Наименование данного справочника. Например:

ВЫБРАТЬ


$ДокСтроки.Товар КАК [Товар $Справочник.Номенклатура]

,$ДокСтроки.Количество КАК [Количество $Число]
ИЗ


ДокументСтроки.ПриходнаяНакладная КАК ДокСтроки $nolock
ЛЕВОЕ СОЕДИНЕНИЕ


Справочник.Номенклатура КАК Спр $nolock
ПО


$ДокСтроки.Товар = $Спр.ТекущийЭлемент
УПОРЯДОЧИТЬ


$ Спр.Наименование ВОЗР
Встроенные функции в языке запросов, операция выбора.
Данные функции могут использоваться для преобразования значений в запросах, могут указываться в качестве полей выборки запроса.
· СРЕД(Выражение,Начальный символ,Конечный символ)

· ЛЕВО(Выражение,Количество символов)

· ПРАВО(Выражение,Количество символов)

· СОКРЛ(Выражение)

· СОКРП(Выражение)

· ГОД(Выражение)

· МЕСЯЦ(Выражение)

· ДЕНЬ(Выражение)

· НАЧАЛОПЕРИОДА(Выражение,[День|Неделя|Месяц|Квартал|Год])

· ОКРУГЛИТЬ(Выражение,Количество знаков)
